eman ta zabal zazu

Universidad del País Vasco
Euskal Herriko Unibertsitatea

Material de estudio

OCW 2018: *Utilizando Mathematica como apoyo al cálculo algebraico en los grados de Ingeniería*

Tema 3. Sistemas de ecuaciones lineales

Equipo docente del curso
Martín Yagüe, Luis
Unzueta Inchaurbe, Aitziber
Arrospide Zabala, Eneko
García Ramiro, María Begoña
Soto Merino, Juan Carlos
Alonso González, Erik

Departamento de Matemática Aplicada
Escuela de Ingeniería de Bilbao, Edificio II-I

TEMA 3. SISTEMAS DE ECUACIONES LINEALES

Sistemas sin parámetros

Funciones

Mathematica resuelve sistemas de ecuaciones lineales con diferentes funciones: **Solve**, **NSolve**, **LinearSolve**.

- **Solve**. Vista en el *Tema 2. Cálculo matricial*.
  - Las ecuaciones del sistema (primer argumento, *expr*) se encierran entre llaves y se separan con comas
  - Las ecuaciones deben estar correctamente definidas (==)
  - Puede definirse el sistema previamente asignándolo a una variable
  - Las variables que se resuelven (segundo argumento, *vars*) se encierran entre llaves y se separan con comas
  - Si el sistema es **compatible indeterminado**: devuelve las infinitas soluciones como único elemento de una lista y presenta un mensaje de aviso que indica que se despejan unas variables en función de otras
  - Si el sistema es **compatible determinado** da la solución como único elemento de una lista
  - Si el sistema es **incompatible** devuelve una lista vacía

Ejemplo. Resolver el sistema: $\begin{cases} x+y+z+t & = 0 \\ x+y-z-t & = 0 \end{cases}$

```
sist1 = {x + y + z + t == 0, x + y - z - t == 0};
```

```
Solve[sist1, {x, y, z, t}] (* sistema compatible indeterminado *)
```

Solve: Equations may not give solutions for all "solve" variables.

```
{{y → -x, t → -z}}
```

Ejemplo. Resolver el sistema: $\begin{cases} x+2\,y & = 4 \\ 3\,x+4\,y & = 6 \end{cases}$

```
sist2 = {x + 2 y == 4, 3 x + 4 y == 6};
```

```
Solve[sist2, {x, y}] (* sistema compatible determinado *)
```

```
{{x → -2, y → 3}}
```

Ejemplo. Resolver el sistema: $\begin{cases} x+2\,y & = 3 \\ 2\,x+4\,y & = 3 \end{cases}$

```
sist3 = {x + 2 y == 3, 2 x + 4 y == 3};
```

```
Solve[sist3, {x, y}] (* sistema incompatible *)
```

```
{}
```

- **LinearSolve**. Resuelve un sistema de ecuaciones dado en su forma matricial.

- Si el sistema es **compatible indeterminado** sólo devuelve una de las infinitas posibles soluciones.

Ejemplo. Resolver el sistema: $\begin{cases} x+y+z+t &= 0 \\ x+y-z-t &= 0 \end{cases} \Longrightarrow \begin{pmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & -1 & -1 \end{pmatrix} \begin{pmatrix} x \\ y \\ z \\ t \end{pmatrix} = \begin{pmatrix} 0 \\ 0 \end{pmatrix}$

```
{ma1 = {{1, 1, 1, 1}, {1, 1, -1, -1}}, b1 = {0, 0}};

LinearSolve[ma1, b1] (* sistema compatible indeterminado *)

{0, 0, 0, 0}
```

Ejemplo. Resolver el sistema: $\begin{cases} x+2y &= 4 \\ 3x+4y &= 6 \end{cases} \Longrightarrow \begin{pmatrix} 1 & 2 \\ 3 & 4 \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 4 \\ 6 \end{pmatrix}$

```
{ma2 = {{1, 2}, {3, 4}}, b2 = {4, 6}};

LinearSolve[ma2, b2] (* sistema compatible determinado *)

{-2, 3}
```

Ejemplo. Resolver el sistema: $\begin{matrix} x+2y &= 3 \\ 2x+4y &= 3 \end{matrix} \Longrightarrow \begin{pmatrix} 1 & 2 \\ 2 & 4 \end{pmatrix} \begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 3 \\ 3 \end{pmatrix}$

```
{ma3 = {{1, 2}, {2, 4}}, b3 = {3, 3}};

LinearSolve[ma3, b3] (* sistema incompatible *)

LinearSolve: Linear equation encountered that has no solution.

LinearSolve[{{1, 2}, {2, 4}}, {3, 3}]
```

- **NSolve**. Análogo a **Solve** aunque obtiene la solución utilizando métodos numéricos de aproximación a la solución.

Ejemplo. Resolver el sistema: $\begin{cases} x+\sqrt{2}\,y &= 5 \\ e\cdot x-\sqrt[3]{4}\,y &= 0 \end{cases}$

$$\text{Solve}\left[\left\{x + \sqrt{2}\ y == 5,\ E*x - \sqrt[3]{4}\ y == 0\right\},\ \{x, y\}\right]\ //\ \text{Simplify}$$

$$\left\{\left\{x \to \frac{5\times 2^{1/6}}{2^{1/6}+e},\ y \to \frac{5\,e}{\sqrt{2}\ \left(2^{1/6}+e\right)}\right\}\right\}$$

$$\text{NSolve}\left[\left\{x + \sqrt{2}\ y == 5,\ E*x - \sqrt[3]{4}\ y == 0\right\},\ \{x, y\}\right]$$

```
{{x → 1.46126, y → 2.50227}}
```

Ejemplo. Resolver el sistema: $\begin{cases} x+\sqrt{2}\ y-ln(4)\,z &= 5 \\ 3x+4y-\left(log_{10}3\right)z &= 6 \\ e\cdot x-4z &= 0 \end{cases}$

$$\text{Solve}\left[\left\{x + \sqrt{2}\ y - \text{Log}[4]\ z == 5,\ 3x + 4y + \text{Log}[10, 3]\ z == 6,\ E*x - 4z == 0\right\},\ \{x, y, z\}\right]\ //\ \text{Simplify}$$

$$\left\{\left\{x \to \left(8\left(-10+3\sqrt{2}\right)\text{Log}[10]\right)\Big/\left(4\left(-4+3\sqrt{2}\right)\text{Log}[10]+e\left(\sqrt{2}\ \text{Log}[3]+4\,\text{Log}[4]\,\text{Log}[10]\right)\right),\right.\right.$$
$$y \to \left(5\,e\,\text{Log}[3]+36\,\text{Log}[10]+6\,e\,\text{Log}[4]\,\text{Log}[10]\right)\Big/\left(4\left(-4+3\sqrt{2}\right)\text{Log}[10]+e\left(\sqrt{2}\ \text{Log}[3]+4\,\text{Log}[4]\,\text{Log}[10]\right)\right),$$
$$\left.\left.z \to \left(2\left(-10+3\sqrt{2}\right)e\,\text{Log}[10]\right)\Big/\left(4\left(-4+3\sqrt{2}\right)\text{Log}[10]+e\left(\sqrt{2}\ \text{Log}[3]+4\,\text{Log}[4]\,\text{Log}[10]\right)\right)\right\}\right\}$$

```
NSolve[{x + √2 y - Log[4] z == 5, 3 x + 4 y + Log[10, 3] z == 6, E * x - 4 z == 0}, {x, y, z}]
```

```
{{x → -2.57628, y → 3.64104, z → -1.75076}}
```

Representación gráfica

Un sistema de ecuaciones lineales de dos incógnitas se interpreta geométricamente como un conjunto de rectas en el plano. La solución del sistema la constituyen sus puntos comunes.

Para la representación gráfica se usan las funciones:

- **`Line`**. Representa los segmentos de recta que unen una secuencia de puntos.
- **`Graphics`**. Representa una imagen gráfica bidimensional. Se recomienda consultar la *Ayuda* para profundizar en el conocimiento de sus *primitivas* y sus *opciones*:

Graphics[*primitivas*,*opciones*]

Ejemplo. Resolver el sistema: $\begin{cases} x+2\,y & =4 \\ 3\,x+4\,y & =6 \end{cases}$

```
sist = {x + 2 y == 4, 3 x + 4 y == 6};
```

```
sol = Solve[sist, {x, y}]
```

```
{{x → -2, y → 3}}
```

```
r1 = Line[{{4, 0}, {-4, 4}}];
```

```
r2 = Line[{{2, 0}, {-4, 9 / 2}}];
```

```
Graphics[{{Lighter[Blue, 0.5], r1, r2}, {Red, PointSize[0.025], Point[{x, y} /. sol]}},
 Axes → True, ImageSize → Small]
```



Ejemplo. Resolver el sistema: $\begin{cases} x+\sqrt{2}\;y & =5 \\ e\cdot x-\sqrt[3]{4}\;y & =0 \end{cases}$

```
sist = {x + √2 y == 5, E * x - ∛4 y == 0};
```

```
sol2 = NSolve[sist, {x, y}]
```

```
{{x → 1.46126, y → 2.50227}}
```

```
r3 = Line[{{0, 5 / Sqrt[2]}, {3, Sqrt[2]}}];
```

```
r4 = Line[{{0, 0}, {3, 3 E / Power[4, 1 / 3]}}];
```

```
Graphics[{{Lighter[Blue, 0.5], r3, r4}, {Red, PointSize[0.035], Point[{x, y} /. sol2]}},
 Axes → True, Ticks → {{0, 1, 1.46}, {0, 1, 2.50}}, ImageSize → Small]
```



Un sistema de ecuaciones lineales de tres incógnitas se interpreta geométricamente como un conjunto de planos en el espacio. La solución del sistema la constituyen sus puntos comunes.

Para la representación gráfica se usan las funciones:

- **InfinitePlane**. Representa el plano que pasa por tres puntos o que pasa por un punto en las direcciones de dos vectores.
- **Graphics3D**. Representa una imagen gráfica tridimensional. Se recomienda acudir a la *Ayuda* para profundizar en el conocimiento de sus *primitivas* y sus *opciones*.

Ejemplo. Resolver y representar gráficamente el sistema: $\begin{cases} x-y &= 0 \\ x-z &= 0 \\ y &= 1 \end{cases}$

```
sist = {x - y == 0, x - z == 0, y == 1};

sol3 = Solve[sist, {x, y, z}] (* los tres planos se cortan en un punto *)

{{x → 1, y → 1, z → 1}}

p1 = InfinitePlane[{{1, 1, 0}, {1, 1, 1}, {0, 0, 1}}]; (* plano que contiene tres puntos *)

p2 = InfinitePlane[{{1, 0, 1}, {1, 1, 1}, {0, 0, 0}}]; (* plano que contiene tres puntos *)

{pin = {1, 1, 1}, v1 = {1, 0, 1}, v2 = {1, 0, -1}};

p3 = InfinitePlane[pin, {v1, v2}]; (* plano que contiene un punto y dos vectores *)

Graphics3D[{Opacity[0.35], LightGreen, p1, p2, p3, {Red, PointSize[0.05], Point[{x, y, z} /. sol3]}},
 Boxed → False, Axes → True, AxesLabel → {"X", "Y", "Z"}, Ticks → {{0, 1, 2}, {0, 1, 2}, {0, 1, 2}},
 BoxRatios → Automatic, ImageSize → Small, PlotRange → {{0, 2}, {0, 2}, {0, 2}}]
```



Sistemas con parámetros

Discusión de un sistema

La función **Reduce** facilita la discusión sobre la compatibilidad de un sistema en el que, además de las incógnitas, aparecen diferentes parámetros reales.

- **Reduce**. Resuelve un sistema en función de los valores de uno o varios parámetros.
  - Las ecuaciones del sistema (primer argumento, *expr*) se encierran entre llaves y se separan con comas
  - Puede definirse el sistema previamente asignándolo a una variable
  - Las ecuaciones deben estar correctamente definidas (==)
  - Las variables que se resuelven (segundo argumento, *vars*) se encierran entre llaves y se separan con comas
  - La salida muestra la resolución de los casos de compatibilidad del sistema
  - Los casos no contemplados en la salida se corresponden con los valores de los parámetros para los que el sistema resulta incompatible
  - En la salida, los diferentes casos se presentan y combinan con la ayuda de los operadores: || (ó lógica) y && (y lógica)

Ejemplo. Resolver el siguiente sistema según los valores de $q \in \mathbb{R}$: $\begin{cases} x+2y &= 4 \\ 3x+qy &= 6 \end{cases}$

```
Reduce[{x + 2 y == 4, 3 x + q * y == 6}, {x, y}]
```

$-6+q \neq 0 \;\&\&\; x == \frac{4(-3+q)}{-6+q} \;\&\&\; y == \frac{4-x}{2}$

```
(4 - x)/2 /. x -> (4 (-3 + q))/(-6 + q) // Simplify
```

$-\frac{6}{-6+q}$

- Discusión y resolución en los casos de compatibilidad

Parámetros	***Naturaleza***	***Solución***
$q \neq 6$	Compatible determinado	$\begin{cases} x = \frac{4(-3+q)}{-6+q} \\ y = -\frac{6}{-6+q} \end{cases}$
$q = 6$	Incompatible	−−

Ejemplo completo de resolución con dos parámetros

Dado el sistema de ecuaciones lineales : $S \equiv \begin{cases} p \cdot x_1 + x_2 + x_3 = 0 \\ -x_1 + p \cdot x_2 - x_3 = 0 \\ x_1 + x_2 + x_3 = q \end{cases}$

a) Clasificar el sistema según los valores de los parámetros reales $p,q \in \mathbb{R}$; resolverlo en los casos de compatibilidad

b) Hallar los rangos de las matrices A y $A|B$, siendo A la matriz de coeficientes y $A|B$ la matriz ampliada del sistema; justificar las respuestas

a) Clasificación y resolución

- Introducción del sistema

```
s = {p*x1 + x2 + x3 == 0, -x1 + p*x2 - x3 == 0, x1 + x2 + x3 == q};
```

- Reducción del sistema

```
Reduce[s, {x1, x2, x3}]
```

$(q == 0 \&\& p == -1 \&\& x1 == 0 \&\& x3 == -x2) \,||\, (q == 0 \&\& p == 1 \&\& x2 == 0 \&\& x3 == -x1) \,||$
$\left(-1 + p \neq 0 \&\& x1 == -\frac{q}{-1+p} \&\& 1 + p \neq 0 \&\& x2 == \frac{q}{1+p} \&\& x3 == q - x1 - x2\right)$

```
q - x1 - x2 /. {x1 -> -q/(-1 + p), x2 -> q/(1 + p)} // Simplify (* cálculo de x3 en el tercer caso *)
```

$$\frac{(1+p^2)\, q}{-1+p^2}$$

- Discusión y resolución en los casos de compatibilidad

Parámetros	***Naturaleza***	***Solución***
$p = -1 \wedge q = 0$	Compatible indeterminado	$\begin{cases} x_1 = 0 \\ x_2 = -x_3 \end{cases} \forall\, x_3 \in \mathbb{R}$
$p = 1 \wedge q = 0$	Compatible indeterminado	$\begin{cases} x_2 = 0 \\ x_1 = -x_3 \end{cases} \forall\, x_3 \in \mathbb{R}$
$(p \neq 1 \wedge p \neq -1)\ \forall\, q$	Compatible determinado	$\begin{cases} x_1 = -\frac{q}{-1+p} \\ x_2 = \frac{q}{1+p} \\ x_3 = \frac{(1+p^2)q}{-1+p^2} \end{cases} \forall\, q \in \mathbb{R}$
$(p = 1 \vee p = -1) \wedge q \neq 0$	Incompatible	$--$

b) Rango de las matrices

- Matriz de coeficientes del sistema

```
a = {{p, 1, 1}, {-1, p, -1}, {1, 1, 1}};
```

- Matriz ampliada del sistema

```
am = {{p, 1, 1, 0}, {-1, p, -1, 0}, {1, 1, 1, q}};
```

- Caso 1: $p = -1 \wedge q = 0$

```
{MatrixRank[a /. {p -> -1, q → 0}], MatrixRank[am /. {p -> -1, q → 0}]}
```

{2, 2}

- Caso 2: $p = 1 \wedge q = 0$

```
{MatrixRank[a /. {p -> 1, q → 0}], MatrixRank[am /. {p -> 1, q → 0}]}
```

{2, 2}

- Caso 3: $p \neq -1 \wedge p \neq 1 \ \ \forall q \in \mathbb{R}$

```
Det[a] (* en este caso: Det[a]≠0 ⇒ r(a)=3 *)
```

$-1 + p^2$

- Caso 4: $q \neq 0 \wedge (p = -1 \vee p = 1)$

```
Det[a] (* en este caso: Det[a]=0 ⇒ r(a)<3 *)
```

$-1 + p^2$

```
la = Flatten[Minors[a, 2]]
```

$\{1 + p^2, 1 - p, -1 - p, -1 + p, -1 + p, 0, -1 - p, 0, 1 + p\}$

```
sol = Solve[Table[la[[i]] == 0, {i, Length[la]}], p] (* no hay solución: r(a)=2 *)
```

{}

```
lam = Flatten[Minors[am, 3]]
```

$\{-1 + p^2, q + p^2 q, q - p q, -q - p q\}$

```
solm = Solve[Table[lam[[i]] == 0, {i, Length[lam]}], {p, q}] (* en este caso: r(am)=3 *)
```

{{p → -1, q → 0}, {p → 1, q → 0}}

Parámetros	***Naturaleza***	***Rangos***
$\boldsymbol{p = -1 \wedge q = 0}$	Compatible indeterminado	$r(A) = r(A \mid B) = 2$
$\boldsymbol{p = 1 \wedge q = 0}$	Compatible indeterminado	$r(A) = r(A \mid B) = 2$
$\boldsymbol{(p \neq 1 \wedge p \neq -1) \ \forall q}$	Compatible determinado	$r(A) = r(A \mid B) = 3$
$\boldsymbol{(p = 1 \vee p = -1) \wedge q \neq 0}$	Incompatible	$r(A) = 2 \neq r(A \mid B) = 3$